Rules for integrands involving error functions

1. $\int \mathrm{Erf}[a + b x]^n \, dx$

   1: $\int \mathrm{Erf}[a + b x] \, dx$

Reference: G&R 5.41

Derivation: Integration by parts

Basis: $\partial_x \mathrm{Erf}[a + b x] == \frac{2 b}{\sqrt{\pi} \, e^{(a+b x)^2}}$

– Rule:

$$\int \mathrm{Erf}[a + b x] \, dx \longrightarrow \frac{(a + b x) \, \mathrm{Erf}[a + b x]}{b} - \frac{2}{\sqrt{\pi}} \int \frac{a + b x}{e^{(a+b x)^2}} \, dx \longrightarrow \frac{(a + b x) \, \mathrm{Erf}[a + b x]}{b} + \frac{1}{b \sqrt{\pi} \, e^{(a+b x)^2}}$$

– Program code:

```
Int[Erf[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*Erf[a+b*x]/b + 1/(b*Sqrt[Pi]*E^(a+b*x)^2) /;
FreeQ[{a,b},x]
```

```
Int[Erfc[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*Erfc[a+b*x]/b - 1/(b*Sqrt[Pi]*E^(a+b*x)^2) /;
FreeQ[{a,b},x]
```

```
Int[Erfi[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*Erfi[a+b*x]/b - E^(a+b*x)^2/(b*Sqrt[Pi]) /;
FreeQ[{a,b},x]
```

2: $\int \text{Erf}[a+b\,x]^2\,dx$

Derivation: Integration by parts

Basis: $\partial_x \text{Erf}[a+b\,x]^2 = \frac{4\,b\,\text{Erf}[a+b\,x]}{\sqrt{\pi}\;e^{(a+b\,x)^2}}$

Rule:

$$\int \text{Erf}[a+b\,x]^2\,dx \to \frac{(a+b\,x)\,\text{Erf}[a+b\,x]^2}{b} - \frac{4}{\sqrt{\pi}}\int \frac{(a+b\,x)\,\text{Erf}[a+b\,x]}{e^{(a+b\,x)^2}}\,dx$$

Program code:

```
Int[Erf[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*Erf[a+b*x]^2/b -
  4/Sqrt[Pi]*Int[(a+b*x)*Erf[a+b*x]/E^(a+b*x)^2,x] /;
FreeQ[{a,b},x]
```

```
Int[Erfc[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*Erfc[a+b*x]^2/b +
  4/Sqrt[Pi]*Int[(a+b*x)*Erfc[a+b*x]/E^(a+b*x)^2,x] /;
FreeQ[{a,b},x]
```

```
Int[Erfi[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*Erfi[a+b*x]^2/b -
  4/Sqrt[Pi]*Int[(a+b*x)*E^(a+b*x)^2*Erfi[a+b*x],x] /;
FreeQ[{a,b},x]
```

U: $\int \text{Erf}[a+b\,x]^n \,dx$ when $n \neq 1 \land n \neq 2$

Rule: If $n \neq 1 \land n \neq 2$, then

$$\int \text{Erf}[a+b\,x]^n \,dx \longrightarrow \int \text{Erf}[a+b\,x]^n \,dx$$

Program code:

```
Int[Erf[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[Erf[a+b*x]^n,x] /;
FreeQ[{a,b,n},x] && NeQ[n,1] && NeQ[n,2]
```

```
Int[Erfc[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[Erfc[a+b*x]^n,x] /;
FreeQ[{a,b,n},x] && NeQ[n,1] && NeQ[n,2]
```

```
Int[Erfi[a_.+b_.*x_]^n_,x_Symbol] :=
  Unintegrable[Erfi[a+b*x]^n,x] /;
FreeQ[{a,b,n},x] && NeQ[n,1] && NeQ[n,2]
```

2. $\int (c + d x)^m \text{Erf}[a + b x]^n \, dx$

1. $\int (c + d x)^m \text{Erf}[a + b x] \, dx$

1: $\int \frac{\text{Erf}[b x]}{x} \, dx$

- Basis: $\text{Erfc}[z] == 1 - \text{Erf}[z]$

- Rule:

$$\int \frac{\text{Erf}[b x]}{x} \, dx \longrightarrow \frac{2 b x}{\sqrt{\pi}} \text{HypergeometricPFQ}\left[\left\{\frac{1}{2}, \frac{1}{2}\right\}, \left\{\frac{3}{2}, \frac{3}{2}\right\}, -b^2 x^2\right]$$

- Program code:

```
Int[Erf[b_.*x_]/x_,x_Symbol] :=
  2*b*x/Sqrt[Pi]*HypergeometricPFQ[{1/2,1/2},{3/2,3/2},-b^2*x^2] /;
FreeQ[b,x]
```

```
Int[Erfc[b_.*x_]/x_,x_Symbol] :=
  Log[x] - Int[Erf[b*x]/x,x] /;
FreeQ[b,x]
```

```
Int[Erfi[b_.*x_]/x_,x_Symbol] :=
  2*b*x/Sqrt[Pi]*HypergeometricPFQ[{1/2,1/2},{3/2,3/2},b^2*x^2] /;
FreeQ[b,x]
```

2: $\int (c+d\,x)^m \,\text{Erf}[a+b\,x]\,dx$ when $m \neq -1$

Derivation: Integration by parts

Basis: $\partial_x \text{Erf}[a+b\,x] == \frac{2\,b}{\sqrt{\pi}\,e^{(a+b\,x)^2}}$

Rule: If $m \neq -1$, then

$$\int (c+d\,x)^m \,\text{Erf}[a+b\,x]\,dx \longrightarrow \frac{(c+d\,x)^{m+1}\,\text{Erf}[a+b\,x]}{d\,(m+1)} - \frac{2\,b}{\sqrt{\pi}\,d\,(m+1)} \int \frac{(c+d\,x)^{m+1}}{e^{(a+b\,x)^2}}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Erf[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*Erf[a+b*x]/(d*(m+1)) -
  2*b/(Sqrt[Pi]*d*(m+1))*Int[(c+d*x)^(m+1)/E^(a+b*x)^2,x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

```
Int[(c_.+d_.*x_)^m_.*Erfc[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*Erfc[a+b*x]/(d*(m+1)) +
  2*b/(Sqrt[Pi]*d*(m+1))*Int[(c+d*x)^(m+1)/E^(a+b*x)^2,x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

```
Int[(c_.+d_.*x_)^m_.*Erfi[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^(m+1)*Erfi[a+b*x]/(d*(m+1)) -
  2*b/(Sqrt[Pi]*d*(m+1))*Int[(c+d*x)^(m+1)*E^(a+b*x)^2,x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[m,-1]
```

2. $\int (c+d\,x)^m\,\text{Erf}[a+b\,x]^2\,dx$

1: $\int x^m\,\text{Erf}[b\,x]^2\,dx$ when $m\in\mathbb{Z}^+ \vee \frac{m+1}{2}\in\mathbb{Z}^-$

Derivation: Integration by parts

Basis: $\partial_x\,\text{Erf}[b\,x]^2 == \frac{4\,b\,\text{Erf}[b\,x]}{\sqrt{\pi}\,e^{b^2 x^2}}$

Rule: If $m\in\mathbb{Z}^+ \vee \frac{m+1}{2}\in\mathbb{Z}^-$, then

$$\int x^m\,\text{Erf}[b\,x]^2\,dx \longrightarrow \frac{x^{m+1}\,\text{Erf}[b\,x]^2}{m+1}-\frac{4\,b}{\sqrt{\pi}\,(m+1)}\int \frac{x^{m+1}\,\text{Erf}[b\,x]}{e^{b^2 x^2}}\,dx$$

Program code:

```
Int[x_^m_.*Erf[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*Erf[b*x]^2/(m+1) -
  4*b/(Sqrt[Pi]*(m+1))*Int[x^(m+1)*E^(-b^2*x^2)*Erf[b*x],x] /;
FreeQ[b,x] && (IGtQ[m,0] || ILtQ[(m+1)/2,0])
```

```
Int[x_^m_.*Erfc[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*Erfc[b*x]^2/(m+1) +
  4*b/(Sqrt[Pi]*(m+1))*Int[x^(m+1)*E^(-b^2*x^2)*Erfc[b*x],x] /;
FreeQ[b,x] && (IGtQ[m,0] || ILtQ[(m+1)/2,0])
```

```
Int[x_^m_.*Erfi[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*Erfi[b*x]^2/(m+1) -
  4*b/(Sqrt[Pi]*(m+1))*Int[x^(m+1)*E^(b^2*x^2)*Erfi[b*x],x] /;
FreeQ[b,x] && (IGtQ[m,0] || ILtQ[(m+1)/2,0])
```

2: $\int (c+d\,x)^m \,\mathrm{Erf}[a+b\,x]^2 \,dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by substitution

–

Rule: If $m \in \mathbb{Z}^+$, then

$$\int (c+d\,x)^m \,\mathrm{Erf}[a+b\,x]^2 \,dx \longrightarrow \frac{1}{b^{m+1}} \,\mathrm{Subst}\Big[\int \mathrm{Erf}[x]^2 \,\mathrm{ExpandIntegrand}\big[(b\,c-a\,d+d\,x)^m,\, x\big] \,dx,\, x,\, a+b\,x\Big]$$

–

Program code:

```
Int[(c_.+d_.*x_)^m_.*Erf[a_+b_.*x_]^2,x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[Erf[x]^2,(b*c-a*d+d*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*Erfc[a_+b_.*x_]^2,x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[Erfc[x]^2,(b*c-a*d+d*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*Erfi[a_+b_.*x_]^2,x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[Erfi[x]^2,(b*c-a*d+d*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,0]
```

U: $\int (c+d\,x)^m\,\mathrm{Erf}[a+b\,x]^n\,dx$

Rule:

$$\int (c+d\,x)^m\,\mathrm{Erf}[a+b\,x]^n\,dx \longrightarrow \int (c+d\,x)^m\,\mathrm{Erf}[a+b\,x]^n\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Erf[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*Erf[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

```
Int[(c_.+d_.*x_)^m_.*Erfc[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*Erfc[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

```
Int[(c_.+d_.*x_)^m_.*Erfi[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(c+d*x)^m*Erfi[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

3. $\int e^{c+d\,x^2}\,\mathrm{Erf}[a+b\,x]^n\,dx$

1. $\int e^{c+d\,x^2}\,\mathrm{Erf}[b\,x]^n\,dx$ when $d^2 == b^4$

1: $\int e^{c+d\,x^2}\,\mathrm{Erf}[b\,x]^n\,dx$ when $d == -b^2$

Derivation: Integration by substitution

Basis: If $d == -b^2$, then $e^{c+d\,x^2}\,F[\mathrm{Erf}[b\,x]] == \frac{e^c\sqrt{\pi}}{2\,b}\,\mathrm{Subst}[F[x],\,x,\,\mathrm{Erf}[b\,x]]\,\partial_x \mathrm{Erf}[b\,x]$

Rule: If $d == -b^2$, then

$$\int e^{c+d x^2} \text{Erf}[b x]^n \, dx \longrightarrow \frac{e^c \sqrt{\pi}}{2 b} \text{Subst}\left[\int x^n \, dx, x, \text{Erf}[b x]\right]$$

Program code:

```
Int[E^(c_.+d_.*x_^2)*Erf[b_.*x_]^n_.,x_Symbol] :=
  E^c*Sqrt[Pi]/(2*b)*Subst[Int[x^n,x],x,Erf[b*x]] /;
FreeQ[{b,c,d,n},x] && EqQ[d,-b^2]
```

```
Int[E^(c_.+d_.*x_^2)*Erfc[b_.*x_]^n_.,x_Symbol] :=
  -E^c*Sqrt[Pi]/(2*b)*Subst[Int[x^n,x],x,Erfc[b*x]] /;
FreeQ[{b,c,d,n},x] && EqQ[d,-b^2]
```

```
Int[E^(c_.+d_.*x_^2)*Erfi[b_.*x_]^n_.,x_Symbol] :=
  E^c*Sqrt[Pi]/(2*b)*Subst[Int[x^n,x],x,Erfi[b*x]] /;
FreeQ[{b,c,d,n},x] && EqQ[d,b^2]
```

2: $\int e^{c+d x^2} \text{Erf}[b x] \, dx$ when $d == b^2$

Basis: $\text{Erfc}[z] == 1 - \text{Erf}[z]$

Rule: If $d == b^2$, then

$$\int e^{c+d x^2} \text{Erf}[b x] \, dx \longrightarrow \frac{b e^c x^2}{\sqrt{\pi}} \text{HypergeometricPFQ}\left[\{1, 1\}, \left\{\frac{3}{2}, 2\right\}, b^2 x^2\right]$$

Program code:

```
Int[E^(c_.+d_.*x_^2)*Erf[b_.*x_],x_Symbol] :=
  b*E^c*x^2/Sqrt[Pi]*HypergeometricPFQ[{1,1},{3/2,2},b^2*x^2] /;
FreeQ[{b,c,d},x] && EqQ[d,b^2]
```

```
Int[E^(c_.+d_.*x_^2)*Erfc[b_.*x_],x_Symbol] :=
  Int[E^(c+d*x^2),x] - Int[E^(c+d*x^2)*Erf[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d,b^2]
```

```
Int[E^(c_.+d_.*x_^2)*Erfi[b_.*x_],x_Symbol] :=
  b*E^c*x^2/Sqrt[Pi]*HypergeometricPFQ[{1,1},{3/2,2},-b^2*x^2] /;
FreeQ[{b,c,d},x] && EqQ[d,-b^2]
```

U: $\int e^{c+d\,x^2}\,\mathrm{Erf}[a+b\,x]^n\,dx$

Rule:

$$\int e^{c+d\,x^2}\,\mathrm{Erf}[a+b\,x]^n\,dx \longrightarrow \int e^{c+d\,x^2}\,\mathrm{Erf}[a+b\,x]^n\,dx$$

Program code:

```
Int[E^(c_.+d_.*x_^2)*Erf[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[E^(c+d*x^2)*Erf[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[E^(c_.+d_.*x_^2)*Erfc[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[E^(c+d*x^2)*Erfc[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[E^(c_.+d_.*x_^2)*Erfi[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[E^(c+d*x^2)*Erfi[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x]
```

4. $\int (e\,x)^m\, e^{c+d\,x^2}\,\text{Erf}[a+b\,x]^n\,dx$

1. $\int x^m\, e^{c+d\,x^2}\,\text{Erf}[a+b\,x]\,dx$ when $m\in\mathbb{Z}$

1. $\int x^m\, e^{c+d\,x^2}\,\text{Erf}[a+b\,x]\,dx$ when $m\in\mathbb{Z}^+$

1: $\int x\, e^{c+d\,x^2}\,\text{Erf}[a+b\,x]\,dx$

Derivation: Integration by parts

Basis: $\int x\, e^{c+d\,x^2}\,dx == \frac{1}{2\,d}\,e^{c+d\,x^2}$

Basis: $\partial_x \text{Erf}[a+b\,x] == \frac{2\,b}{\sqrt{\pi}}\,e^{-a^2-2\,a\,b\,x-b^2\,x^2}$

Rule:

$$\int x\, e^{c+d\,x^2}\,\text{Erf}[a+b\,x]\,dx \longrightarrow \frac{e^{c+d\,x^2}\,\text{Erf}[a+b\,x]}{2\,d} - \frac{b}{d\,\sqrt{\pi}}\int e^{-a^2+c-2\,a\,b\,x-\left(b^2-d\right)\,x^2}\,dx$$

Program code:

```
Int[x_*E^(c_.+d_.*x_^2)*Erf[a_.+b_.*x_],x_Symbol] :=
  E^(c+d*x^2)*Erf[a+b*x]/(2*d) -
  b/(d*Sqrt[Pi])*Int[E^(-a^2+c-2*a*b*x-(b^2-d)*x^2),x] /;
FreeQ[{a,b,c,d},x]
```

```
Int[x_*E^(c_.+d_.*x_^2)*Erfc[a_.+b_.*x_],x_Symbol] :=
  E^(c+d*x^2)*Erfc[a+b*x]/(2*d) +
  b/(d*Sqrt[Pi])*Int[E^(-a^2+c-2*a*b*x-(b^2-d)*x^2),x] /;
FreeQ[{a,b,c,d},x]
```

```
Int[x_*E^(c_.+d_.*x_^2)*Erfi[a_.+b_.*x_],x_Symbol] :=
  E^(c+d*x^2)*Erfi[a+b*x]/(2*d) -
  b/(d*Sqrt[Pi])*Int[E^(a^2+c+2*a*b*x+(b^2+d)*x^2),x] /;
FreeQ[{a,b,c,d},x]
```

2: $\int x^m e^{c+d x^2} \operatorname{Erf}[a+b x] \, dx$ when $m-1 \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\int x \, e^{c+d x^2} \, dx = \frac{1}{2 d} e^{c+d x^2}$

Basis: $\partial_x \left(x^{m-1} \operatorname{Erf}[a+b x]\right) = \frac{2 b}{\sqrt{\pi}} x^{m-1} e^{-a^2-2 a b x-b^2 x^2} + (m-1) x^{m-2} \operatorname{Erf}[a+b x]$

Rule: If $m-1 \in \mathbb{Z}^+$, then

$$\int x^m e^{c+d x^2} \operatorname{Erf}[a+b x] \, dx \longrightarrow$$

$$\frac{x^{m-1} e^{c+d x^2} \operatorname{Erf}[a+b x]}{2 d} - \frac{b}{d \sqrt{\pi}} \int x^{m-1} e^{-a^2+c-2 a b x-\left(b^2-d\right) x^2} \, dx - \frac{m-1}{2 d} \int x^{m-2} e^{c+d x^2} \operatorname{Erf}[a+b x] \, dx$$

Program code:

```
Int[x_^m_*E^(c_.+d_.*x_^2)*Erf[a_.+b_.*x_],x_Symbol] :=
  x^(m-1)*E^(c+d*x^2)*Erf[a+b*x]/(2*d) -
  b/(d*Sqrt[Pi])*Int[x^(m-1)*E^(-a^2+c-2*a*b*x-(b^2-d)*x^2),x] -
  (m-1)/(2*d)*Int[x^(m-2)*E^(c+d*x^2)*Erf[a+b*x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,1]
```

```
Int[x_^m_*E^(c_.+d_.*x_^2)*Erfc[a_.+b_.*x_],x_Symbol] :=
  x^(m-1)*E^(c+d*x^2)*Erfc[a+b*x]/(2*d) +
  b/(d*Sqrt[Pi])*Int[x^(m-1)*E^(-a^2+c-2*a*b*x-(b^2-d)*x^2),x] -
  (m-1)/(2*d)*Int[x^(m-2)*E^(c+d*x^2)*Erfc[a+b*x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,1]
```

```
Int[x_^m_*E^(c_.+d_.*x_^2)*Erfi[a_.+b_.*x_],x_Symbol] :=
  x^(m-1)*E^(c+d*x^2)*Erfi[a+b*x]/(2*d) -
  b/(d*Sqrt[Pi])*Int[x^(m-1)*E^(a^2+c+2*a*b*x+(b^2+d)*x^2),x] -
  (m-1)/(2*d)*Int[x^(m-2)*E^(c+d*x^2)*Erfi[a+b*x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[m,1]
```

2. $\int x^m e^{c+d x^2} \operatorname{Erf}[a+b x] \, dx$ when $m \in \mathbb{Z}^-$

1: $\int \frac{e^{c+d x^2} \operatorname{Erf}[b x]}{x} \, dx$ when $d == b^2$

Basis: $\operatorname{Erfc}[z] == 1 - \operatorname{Erf}[z]$

Rule: If $d == b^2$, then

$$\int \frac{e^{c+d x^2} \operatorname{Erf}[b x]}{x} \, dx \longrightarrow \frac{2 b e^c x}{\sqrt{\pi}} \operatorname{HypergeometricPFQ}\left[\left\{\frac{1}{2}, 1\right\}, \left\{\frac{3}{2}, \frac{3}{2}\right\}, b^2 x^2\right]$$

Program code:

```
Int[E^(c_.+d_.*x_^2)*Erf[b_.*x_]/x_,x_Symbol] :=
  2*b*E^c*x/Sqrt[Pi]*HypergeometricPFQ[{1/2,1},{3/2,3/2},b^2*x^2] /;
FreeQ[{b,c,d},x] && EqQ[d,b^2]
```

```
Int[E^(c_.+d_.*x_^2)*Erfc[b_.*x_]/x_,x_Symbol] :=
  Int[E^(c+d*x^2)/x,x] - Int[E^(c+d*x^2)*Erf[b*x]/x,x] /;
FreeQ[{b,c,d},x] && EqQ[d,b^2]
```

```
Int[E^(c_.+d_.*x_^2)*Erfi[b_.*x_]/x_,x_Symbol] :=
  2*b*E^c*x/Sqrt[Pi]*HypergeometricPFQ[{1/2,1},{3/2,3/2},-b^2*x^2] /;
FreeQ[{b,c,d},x] && EqQ[d,-b^2]
```

2: $\int x^m e^{c+d x^2} \operatorname{Erf}[a+b x] \, dx$ when $m+1 \in \mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If $m+1 \in \mathbb{Z}^-$, then

$$\int x^m e^{c+d x^2} \operatorname{Erf}[a+b x] \, dx \longrightarrow$$

$$\frac{x^{m+1}\, e^{c+d\,x^2}\,\text{Erf}[a+b\,x]}{m+1}-\frac{2\,b}{(m+1)\,\sqrt{\pi}}\int x^{m+1}\,e^{-a^2+c-2\,a\,b\,x-\left(b^2-d\right)\,x^2}\,dx-\frac{2\,d}{m+1}\int x^{m+2}\,e^{c+d\,x^2}\,\text{Erf}[a+b\,x]\,dx$$

Program code:

```
Int[x_^m_*E^(c_.+d_.*x_^2)*Erf[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*E^(c+d*x^2)*Erf[a+b*x]/(m+1) -
  2*b/((m+1)*Sqrt[Pi])*Int[x^(m+1)*E^(-a^2+c-2*a*b*x-(b^2-d)*x^2),x] -
  2*d/(m+1)*Int[x^(m+2)*E^(c+d*x^2)*Erf[a+b*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[m,-1]
```

```
Int[x_^m_*E^(c_.+d_.*x_^2)*Erfc[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*E^(c+d*x^2)*Erfc[a+b*x]/(m+1) +
  2*b/((m+1)*Sqrt[Pi])*Int[x^(m+1)*E^(-a^2+c-2*a*b*x-(b^2-d)*x^2),x] -
  2*d/(m+1)*Int[x^(m+2)*E^(c+d*x^2)*Erfc[a+b*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[m,-1]
```

```
Int[x_^m_*E^(c_.+d_.*x_^2)*Erfi[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*E^(c+d*x^2)*Erfi[a+b*x]/(m+1) -
  2*b/((m+1)*Sqrt[Pi])*Int[x^(m+1)*E^(a^2+c+2*a*b*x+(b^2+d)*x^2),x] -
  2*d/(m+1)*Int[x^(m+2)*E^(c+d*x^2)*Erfi[a+b*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[m,-1]
```

U: $\int (e\,x)^m\,e^{c+d\,x^2}\,\text{Erf}[a+b\,x]^n\,dx$

Rule:

$$\int (e\,x)^m\,e^{c+d\,x^2}\,\text{Erf}[a+b\,x]^n\,dx \longrightarrow \int (e\,x)^m\,e^{c+d\,x^2}\,\text{Erf}[a+b\,x]^n\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*E^(c_.+d_.*x_^2)*Erf[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(e*x)^m*E^(c+d*x^2)*Erf[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*E^(c_.+d_.*x_^2)*Erfc[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(e*x)^m*E^(c+d*x^2)*Erfc[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(e_.*x_)^m_.*E^(c_.+d_.*x_^2)*Erfi[a_.+b_.*x_]^n_.,x_Symbol] :=
  Unintegrable[(e*x)^m*E^(c+d*x^2)*Erfi[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

5. $\int u\,\mathrm{Erf}[d\,(a+b\,\mathrm{Log}[c\,x^n])]\,dx$

1: $\int \mathrm{Erf}[d\,(a+b\,\mathrm{Log}[c\,x^n])]\,dx$

Derivation: Integration by parts

Basis: $\partial_x \mathrm{Erf}[d\,(a+b\,\mathrm{Log}[c\,x^n])] == \frac{2\,b\,d\,n}{\sqrt{\pi}\,x\,e^{(d\,(a+b\,\mathrm{Log}[c\,x^n]))^2}}$

Rule:

$$\int \mathrm{Erf}[d\,(a+b\,\mathrm{Log}[c\,x^n])]\,dx \longrightarrow x\,\mathrm{Erf}[d\,(a+b\,\mathrm{Log}[c\,x^n])] - \frac{2\,b\,d\,n}{\sqrt{\pi}}\int \frac{1}{e^{(d\,(a+b\,\mathrm{Log}[c\,x^n]))^2}}\,dx$$

Program code:

```
Int[Erf[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*Erf[d*(a+b*Log[c*x^n])] - 2*b*d*n/(Sqrt[Pi])*Int[1/E^(d*(a+b*Log[c*x^n]))^2,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[Erfc[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*Erfc[d*(a+b*Log[c*x^n])] + 2*b*d*n/(Sqrt[Pi])*Int[1/E^(d*(a+b*Log[c*x^n]))^2,x] /;
FreeQ[{a,b,c,d,n},x]
```

```
Int[Erfi[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*Erfi[d*(a+b*Log[c*x^n])] - 2*b*d*n/(Sqrt[Pi])*Int[E^(d*(a+b*Log[c*x^n]))^2,x] /;
FreeQ[{a,b,c,d,n},x]
```

2: $\int \frac{\mathrm{Erf}\left[d \left(a+b \,\mathrm{Log}\left[c\, x^n\right]\right)\right]}{x} \, dx$

Derivation: Integration by substitution

Basis: $\frac{F[\mathrm{Log}[c\, x^n]]}{x} = \frac{1}{n} \,\mathrm{Subst}[F[x],\, x,\, \mathrm{Log}[c\, x^n]]\; \partial_x \mathrm{Log}[c\, x^n]$

–

Rule:

$$\int \frac{\mathrm{Erf}\left[d \left(a+b \,\mathrm{Log}\left[c\, x^n\right]\right)\right]}{x} \, dx \longrightarrow \frac{1}{n} \,\mathrm{Subst}\left[\mathrm{Erf}[d\,(a+b\,x)],\, x,\, \mathrm{Log}\left[c\, x^n\right]\right]$$

–

Program code:

```
Int[F_[d_.*(a_.+b_.*Log[c_.*x_^n_.])]/x_,x_Symbol] :=
  1/n*Subst[F[d*(a+b*x)],x,Log[c*x^n]] /;
FreeQ[{a,b,c,d,n},x] && MemberQ[{Erf,Erfc,Erfi},F]
```

3: $\int (e\,x)^m\,\mathrm{Erf}[d\,(a+b\,\mathrm{Log}[c\,x^n])]\,dx$ when $m\neq -1$

Derivation: Integration by parts

Basis: $\partial_x\,\mathrm{Erf}[d\,(a+b\,\mathrm{Log}[c\,x^n])] == \frac{2\,b\,d\,n}{\sqrt{\pi}\,x\,e^{(d\,(a+b\,\mathrm{Log}[c\,x^n]))^2}}$

Rule: If $m\neq -1$, then

$$\int (e\,x)^m\,\mathrm{Erf}[d\,(a+b\,\mathrm{Log}[c\,x^n])]\,dx\;\rightarrow\;\frac{(e\,x)^{m+1}\,\mathrm{Erf}[d\,(a+b\,\mathrm{Log}[c\,x^n])]}{e\,(m+1)}-\frac{2\,b\,d\,n}{\sqrt{\pi}\,(m+1)}\int\frac{(e\,x)^m}{e^{(d\,(a+b\,\mathrm{Log}[c\,x^n]))^2}}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Erf[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*Erf[d*(a+b*Log[c*x^n])]/(e*(m+1)) -
  2*b*d*n/(Sqrt[Pi]*(m+1))*Int[(e*x)^m/E^(d*(a+b*Log[c*x^n]))^2,x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

```
Int[(e_.*x_)^m_.*Erfc[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*Erfc[d*(a+b*Log[c*x^n])]/(e*(m+1)) +
  2*b*d*n/(Sqrt[Pi]*(m+1))*Int[(e*x)^m/E^(d*(a+b*Log[c*x^n]))^2,x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

```
Int[(e_.*x_)^m_.*Erfi[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (e*x)^(m+1)*Erfi[d*(a+b*Log[c*x^n])]/(e*(m+1)) -
  2*b*d*n/(Sqrt[Pi]*(m+1))*Int[(e*x)^m*E^(d*(a+b*Log[c*x^n]))^2,x] /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[m,-1]
```

6: $\int \text{Sin}[c + d x^2]\, \text{Erf}[b x]\, dx$ when $d^2 == -b^4$

- Derivation: Algebraic expansion

- Basis: $\sin[c + d x^2] == \frac{1}{2} i\, e^{-i c - i d x^2} - \frac{1}{2} i\, e^{i c + i d x^2}$

  Rule: If $d^2 == -b^4$, then

$$\int \text{Sin}[c + d x^2]\, \text{Erf}[b x]\, dx \longrightarrow \frac{i}{2} \int e^{-i c - i d x^2}\, \text{Erf}[b x]\, dx - \frac{i}{2} \int e^{i c + i d x^2}\, \text{Erf}[b x]\, dx$$

- Program code:

```
Int[Sin[c_.+d_.*x_^2]*Erf[b_.*x_],x_Symbol] :=
  I/2*Int[E^(-I*c-I*d*x^2)*Erf[b*x],x] - I/2*Int[E^(I*c+I*d*x^2)*Erf[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-b^4]
```

```
Int[Sin[c_.+d_.*x_^2]*Erfc[b_.*x_],x_Symbol] :=
  I/2*Int[E^(-I*c-I*d*x^2)*Erfc[b*x],x] - I/2*Int[E^(I*c+I*d*x^2)*Erfc[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-b^4]
```

```
Int[Sin[c_.+d_.*x_^2]*Erfi[b_.*x_],x_Symbol] :=
  I/2*Int[E^(-I*c-I*d*x^2)*Erfi[b*x],x] - I/2*Int[E^(I*c+I*d*x^2)*Erfi[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-b^4]
```

```
Int[Cos[c_.+d_.*x_^2]*Erf[b_.*x_],x_Symbol] :=
  1/2*Int[E^(-I*c-I*d*x^2)*Erf[b*x],x] + 1/2*Int[E^(I*c+I*d*x^2)*Erf[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-b^4]
```

```
Int[Cos[c_.+d_.*x_^2]*Erfc[b_.*x_],x_Symbol] :=
  1/2*Int[E^(-I*c-I*d*x^2)*Erfc[b*x],x] + 1/2*Int[E^(I*c+I*d*x^2)*Erfc[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-b^4]
```

```
Int[Cos[c_.+d_.*x_^2]*Erfi[b_.*x_],x_Symbol] :=
  1/2*Int[E^(-I*c-I*d*x^2)*Erfi[b*x],x] + 1/2*Int[E^(I*c+I*d*x^2)*Erfi[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,-b^4]
```

7: $\int \text{Sinh}[c + d\,x]\ \text{Erf}[b\,x]\,dx$ when $d^2 == b^4$

Derivation: Algebraic expansion

Basis: $\sinh[c + d\,x^2] == \frac{1}{2}\,e^{c+d\,x^2} - \frac{1}{2}\,e^{-c-d\,x^2}$

Rule: If $d^2 == b^4$, then

$$\int \text{Sinh}[c + d\,x^2]\ \text{Erf}[b\,x]\,dx \longrightarrow \frac{1}{2}\int e^{c+d\,x^2}\ \text{Erf}[b\,x]\,dx - \frac{1}{2}\int e^{-c-d\,x^2}\ \text{Erf}[b\,x]\,dx$$

Program code:

```
Int[Sinh[c_.+d_.*x_^2]*Erf[b_.*x_],x_Symbol] :=
  1/2*Int[E^(c+d*x^2)*Erf[b*x],x] - 1/2*Int[E^(-c-d*x^2)*Erf[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,b^4]
```

```
Int[Sinh[c_.+d_.*x_^2]*Erfc[b_.*x_],x_Symbol] :=
  1/2*Int[E^(c+d*x^2)*Erfc[b*x],x] - 1/2*Int[E^(-c-d*x^2)*Erfc[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,b^4]
```

```
Int[Sinh[c_.+d_.*x_^2]*Erfi[b_.*x_],x_Symbol] :=
  1/2*Int[E^(c+d*x^2)*Erfi[b*x],x] - 1/2*Int[E^(-c-d*x^2)*Erfi[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,b^4]
```

```
Int[Cosh[c_.+d_.*x_^2]*Erf[b_.*x_],x_Symbol] :=
  1/2*Int[E^(c+d*x^2)*Erf[b*x],x] + 1/2*Int[E^(-c-d*x^2)*Erf[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,b^4]
```

```
Int[Cosh[c_.+d_.*x_^2]*Erfc[b_.*x_],x_Symbol] :=
  1/2*Int[E^(c+d*x^2)*Erfc[b*x],x] + 1/2*Int[E^(-c-d*x^2)*Erfc[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,b^4]
```

```
Int[Cosh[c_.+d_.*x_^2]*Erfi[b_.*x_],x_Symbol] :=
  1/2*Int[E^(c+d*x^2)*Erfi[b*x],x] + 1/2*Int[E^(-c-d*x^2)*Erfi[b*x],x] /;
FreeQ[{b,c,d},x] && EqQ[d^2,b^4]
```

Rules for integrands involving special functions

1: $\int F[f\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])]\,dx$ when $F\in\{$Erf, Erfc, Erfi, FresnelS, FresnelC, ExpIntegralEi, SinIntegral, CosIntegral, SinhIntegral, CoshIntegral$\}$

Derivation: Integration by substitution

Rule: If $F\in\{$Erf, Erfc, Erfi, FresnelS, FresnelC, ExpIntegralEi, SinIntegral, CosIntegral, SinhIntegral, CoshIntegral$\}$, then

$$\int F[f\,(a+b\,\mathrm{Log}[c\,(d+e\,x)^n])]\,dx\;\longrightarrow\;\frac{1}{e}\,\mathrm{Subst}\Big[\int F[f\,(a+b\,\mathrm{Log}[c\,x^n])]\,dx,\;x,\;d+e\,x\Big]$$

Program code:

```
Int[F_[f_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])],x_Symbol] :=
  1/e*Subst[Int[F[f*(a+b*Log[c*x^n])],x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,n},x] && MemberQ[{Erf,Erfc,Erfi,FresnelS,FresnelC,ExpIntegralEi,SinIntegral,CosIntegral,SinhIntegral,CoshIntegral},F]
```

2: $\int (g+h\,x)^m\,F\big[f\,\big(a+b\,\text{Log}\big[c\,(d+e\,x)^n\big]\big)\big]\,dx$ when
$e\,g-d\,h=0\,\land\,F\in\{\text{Erf},\,\text{Erfc},\,\text{Erfi},\,\text{FresnelS},\,\text{FresnelC},\,\text{ExpIntegralEi},\,\text{SinIntegral},\,\text{CosIntegral},\,\text{SinhIntegral},\,\text{CoshIntegral}\}$

Derivation: Integration by substitution

Basis: If $e\,g-d\,h=0$, then $(g+h\,x)^m\,F[d+e\,x]=\frac{1}{e}\,\text{Subst}\Big[\left(\frac{g\,x}{d}\right)^m\,F[x],\,x,\,d+e\,x\Big]\,\partial_x\,(d+e\,x)$

Rule: If $e\,g-d\,h=0\,\land\,F\in\{\text{Erf},\,\text{Erfc},\,\text{Erfi},\,\text{FresnelS},\,\text{FresnelC},\,\text{ExpIntegralEi},\,\text{SinIntegral},\,\text{CosIntegral},\,\text{SinhIntegral},\,\text{CoshIntegral}\}$, then

$$\int (g+h\,x)^m\,F\big[f\,\big(a+b\,\text{Log}\big[c\,(d+e\,x)^n\big]\big)\big]\,dx\;\longrightarrow\;\frac{1}{e}\,\text{Subst}\Big[\int\left(\frac{g\,x}{d}\right)^m\,F\big[f\,\big(a+b\,\text{Log}\big[c\,x^n\big]\big)\big]\,dx,\,x,\,d+e\,x\Big]$$

Program code:

```
Int[(g_+h_.x_)^m_.*F_[f_.*(a_.+b_.*Log[c_.*(d_+e_.*x_)^n_.])],x_Symbol] :=
  1/e*Subst[Int[(g*x/d)^m*F[f*(a+b*Log[c*x^n])],x],x,d+e*x] /;
FreeQ[{a,b,c,d,e,f,g,m,n},x] && EqQ[e*f-d*g,0] &&
  MemberQ[{Erf,Erfc,Erfi,FresnelS,FresnelC,ExpIntegralEi,SinIntegral,CosIntegral,SinhIntegral,CoshIntegral},F]
```